



プログラミング演習(12)

マルチメディア

中村、小林、三武、辻野



- 同じ所：プログラミング言語なので変数、計算、条件分岐、繰り返し、メソッド、クラスなど同じ

12345の約数の数を数えるプログラム

```
int i = 1;
int count = 0;
while( i <= 12345 ){
    if( (12345 % i) == 0 ){
        // 12345をiで割った余りが
        // 0だったらcountを増やす
        count++;
    }
    i++;
}
println( "約数の数は"+count );
```

```
class Calc {
    public static void main(String[] args){
        int i = 1;
        int count = 0;
        while( i <= 12345 ){
            if( (12345 % i) == 0 ){
                // 12345をiで割った余りが
                // 0だったらcountを増やす
                count++;
            }
            i++;
        }
        System.out.println( "約数の数は"+count );
    }
}
```



- 同じ所：プログラミング言語なので変数、計算、条件分岐、繰り返し、メソッド、クラスなど同じ

12345の約数の数を数えるプログラム

```
int i = 1;
int count = 0;
while( i <= 12345 ){
    if( (12345 % i) == 0 ){
        // 12345をiで割った余りが
        // 0だったらcountを増やす
        count++;
    }
    i++;
}
println( "約数の数は"+count );
```

```
<?php
$i = 1;
$count = 0;
while( $i <= 12345 ){
    if( (12345 % $i) == 0 ){
        // 12345を$iで割った余りが
        // 0だったら$countを増やす
        $count++;
    }
    $i++;
}
echo "約数の数は".$count;
?>
```



- 同じ所：プログラミング言語なので変数、計算、条件分岐、繰り返し、メソッド、クラスなど同じ

12345の約数の数を数えるプログラム

```
int i = 1;
int count = 0;
while( i <= 12345 ){
  if( (12345 % i) == 0 ){
    // 12345をiで割った余りが
    // 0だったらcountを増やす
    count++;
  }
  i++;
}
println( "約数の数は"+count );
```

```
<script>
let i = 1;
let count = 0;
while( i <= 12345 ){
  if( (12345 % i) == 0 ){
    // 12345をiで割った余りが
    // 0だったらcountを増やす
    count++;
  }
  i++;
}
alert( "約数の数は"+count );
</script>
```



- 同じ所: プログラミング言語なので変数、計算、条件分岐、繰り返し、メソッド、クラスなど同じ

12345の約数の数を数えるプログラム

```
int i = 1;
int count = 0;
while( i <= 12345 ){
    if( (12345 % i) == 0 ){
        // 12345をiで割った余りが
        // 0だったらcountを増やす
        count++;
    }
    i++;
}
println( "約数の数は"+count );
```

```
i = 1
count = 0
while i <= 12345:
    if (12345 % i) == 0:
        # 12345をiで割った余りが
        # 0だったらcountを増やす
        count = count + 1
    i = i + 1

print( "約数の数は" + str(count) )
```

Pythonモード

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



sketch_210712b | Processing 3.5.4

ファイル 編集 スケッチ ツール Help

Python ▾

```
sketch_210712b ▾
1 def setup():
2     size(400, 400)
3
4 def draw():
5     background(255, 255, 255)
6     ellipse(mouseX, mouseY, 50, 50)
7
8
9
10
11
```



- Processing で画像や音楽を扱う
 - 画像を表示する
 - 音楽を再生する
 - 効果音を再生する

画像の表示



PImage 画像用変数;

画像用変数 = loadImage("画像名"); で準備

image(画像用変数, x座標, y座標); で表示する

画像はプログラムにドロップで利用可能に

(ドラッグアンドドロップしないと使えない)

```
PImage mapImage = loadImage("map.png");  
size(640, 400);  
background(255, 255, 255);  
image(mapImage, 0, 0);
```



- 場所だけを指定した画像の描画
 - `image(画像用変数, X座標, Y座標);`
 - サイズは画像自体の大きさになる
- 場所とサイズを指定した画像の描画
 - `image(画像用変数, X座標, Y座標, 横幅, 縦幅);`
 - 拡大縮小やゆがめた描画も可能

カーソルの場所に画像を表示



(Q) 画像をカーソルの下に常に表示するには
どうするか？

• 考え方

- 画像を用意して、プログラムにドロップ！
- PImage 型の変数を作成 (cursorImage など)
- loadImage で画像を読み込む
 - cursorImage = loadImage("gazo.jpg");
- draw の度に背景を塗りつぶす
- draw の度に image でマウス位置に画像を表示
 - image(cursorImage, mouseX, mouseY);

カーソルの場所に画像を表示

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



```
PImage cursorImage;  
  
void setup()  
{  
  size(800, 600);  
  cursorImage = loadImage("gazo.jpg");  
}  
  
void draw()  
{  
  background(255);  
  image(cursorImage, mouseX, mouseY);  
}
```

パラパラアニメーション

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



(Q) 用意した10枚の画像をパラパラ切り替えるアニメーションを作りたい



- 考え方

- 画像を10枚用意（ペイントで描いても、写真を撮影してもOK. 名前は適当に順番を付けましょう）
- 要素数が10の PImage 型の配列を作る
 - PImage [] parapara = new PImage [10];
- setup で画像をすべて読み込む（loadImage）
 - parapara[0] = loadImage("gazo0.jpg");
- draw で表示する画像番号を変数 i として準備
- draw の度に i 番目の画像を表示
- i が10になったら0に戻す
- frameRate(10); で draw の更新速度を設定

パラパラアニメーション

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



```
PImage [] parapara = new PImage [10];
int i = 0;

void setup()
{
  size(800, 600);
  parapara[0] = loadImage("gazo0.jpg");
  parapara[1] = loadImage("gazo1.jpg");
  parapara[2] = loadImage("gazo2.jpg");
  parapara[3] = loadImage("gazo3.jpg");
  parapara[4] = loadImage("gazo4.jpg");
  parapara[5] = loadImage("gazo5.jpg");
  parapara[6] = loadImage("gazo6.jpg");
  parapara[7] = loadImage("gazo7.jpg");
  parapara[8] = loadImage("gazo8.jpg");
  parapara[9] = loadImage("gazo9.jpg");
  frameRate(10);
}
```

```
void draw()
{
  image(parapara[i], 0, 0);
  i++;
  if(i==10)
  {
    i=0;
  }
}
```

ちなみに



```
parapara[0] = loadImage("gazo0.jpg");  
parapara[1] = loadImage("gazo1.jpg");  
parapara[2] = loadImage("gazo2.jpg");  
parapara[3] = loadImage("gazo3.jpg");  
parapara[4] = loadImage("gazo4.jpg");  
:  
parapara[9] = loadImage("gazo9.jpg");
```




```
for( int j = 0; j < 10; j++)  
{  
    parapara[j] = loadImage("gazo" + j + ".jpg");  
}
```

ちなみに



```
parapara[0] = loadImage("gazo0.jpg");  
parapara[1] = loadImage("gazo1.jpg");  
parapara[2] = loadImage("gazo2.jpg");  
parapara[3] = loadImage("gazo3.jpg");  
parapara[4] = loadImage("gazo4.jpg");  
:  
parapara[9] = loadImage("gazo9.jpg");
```





```
int j = 0;  
while(j < 10)  
{  
    parapara[j] = loadImage("gazo" + j + ".jpg");  
    j++;  
}
```



- ある程度小さな画像が、画面の端で衝突して跳ね返るプログラムを作ろう
 - 画像の x , y 座標と速度 v_x , v_y を用意し、`draw` の度に座標を変更し、端で跳ね返る
- 画像を利用して、占いをするプログラムを作ろう
 - 大吉、中吉、小吉、大凶の画像を用意する
- マウスカーソルの後を10個の画像が追尾するプログラムを作ろう



- Soundライブラリを利用
- 「準備と再生」「終了」が必須
- Processing の画面にファイルをドラッグアンドドロップする！（dataフォルダに保存される）



ライブラリを入れよう

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



The screenshot shows the 'Contribution Manager' window with the 'Libraries' tab selected. A search bar at the top contains the text 'Sound'. Below the search bar, a table lists various audio-related libraries. The 'Sound' library by 'The Processing Foundation' is highlighted. At the bottom of the window, a detailed view of the 'Sound' library is shown, including its version (2.3.1) and a description: 'Provides a simple way to work with audio.' Action buttons for 'Install', 'Update', and 'Remove' are visible.

Status	Name ↓	Author
	Beads A library for adding flexible realtime audio to P...	Ollie Bown, Benito Crawford, Be...
	Cassette Implementation of Processing Sound APIs f...	Shlomi Hod
	ComposingForEveryone ComposingForEveryone giv...	Guido Kramann
✓	Minim An audio library that provides easy to use clas...	Damien Di Fede and Anderson ...
	oscP5 An Open Sound Control implementation.	Andreas Schlegel
✓	Sound Provides a simple way to work with audio.	The Processing Foundation
	SoundCloud Unofficial Java library, which simplifies t...	Darius Morawiec
	SuperCollider client for Processing Framework to int...	Daniel Jones
	wellen wellen is a framework for exploring and teachi...	Dennis P Paul

Sound 2.3.1
The Processing Foundation

Provides a simple way to work with audio.

↓ Install

↻ Update

× Remove



- 準備：グローバル & setup()
 - 音楽再生ライブラリ（便利関数群）の読み込み
 - `import processing.sound.*;`
 - `SoundFile` にBGMを読み込む
 - `SoundFile bgm = new SoundFile(this, "ファイル名");`
 - `bgm.loop();` // 繰り返し再生
 - `bgm.stop();` // 停止
- `SoundFile` に効果音を読み込む
 - `SoundFile se = new SoundFile(this, "ファイル名");`
 - `se.play();` // 効果音の再生

音楽の再生



```
import processing.sound.*;
SoundFile bgm; //BGM格納用の変数
int x = 0;
int vx = 4;

void setup()
{
  size(400, 400);
  bgm = new SoundFile(this, "bgm.mp3"); //mp3ファイルを指定する
  bgm.loop(); //bgmを再生
}

void draw()
{
  background(0, 0, 0);
  x = x + vx;
  if(x >= width)
  {
    x = width;
    vx = -vx;
  }
  else if(x <= 0)
  {
    x = 0;
    vx = -vx;
  }
  circle(x, 200, 20);
}
```

効果音の再生



```
import processing.sound.*;
SoundFile crash; //衝突サウンド格納用の変数
int x = 0;
int vx = 4;

void setup()
{
  size(400, 400);
  crash = new SoundFile(this, "crash.mp3"); //mp3ファイルを指定する
}

void draw()
{
  background(0, 0, 0);
  x = x + vx;
  if(x >= width)
  {
    x = width;
    vx = -vx;
    crash.play(); //再生
  }
  else if(x <= 0)
  {
    x = 0;
    vx = -vx;
    crash.play(); //再生
  }
  circle(x, 200, 20);
}
```

予習問題



- 2つの移動する円を用意し、円が壁に衝突する度に衝突音が鳴るようにせよ
- 2つの移動する円を用意し、円が壁に衝突する度にそれぞれ違う音が鳴るようにせよ
- 2つの移動する円について、マウスでクリックすると破裂音が鳴るようにせよ